#-------------------------------------------------
#
# Project created by QtCreator 2013-10-04T20:18:59
#
#-------------------------------------------------

QT += network
isEqual(QT_MAJOR_VERSION, 5): QT += widgets

isEqual(QT_MAJOR_VERSION, 4): DEFINES += "QT4"
isEqual(QT_MAJOR_VERSION, 5): DEFINES += "QT5"

VERSION = 1.0.0.7

DEPLOYMENT.display_name = Keppe Gui Framework

shared: win32: TARGET = libkguifw
static: win32: TARGET = kguifw
unix: TARGET = kguifw
TEMPLATE = lib

DEFINES += KGUIFW_LIBRARY

SOURCES += \
    KGuiFw.cpp \
    KPTCheckBox.cpp \
    KPTFileDialog.cpp \
    KPTFrame.cpp \
    KPTLabel.cpp \
    KPTLineEdit.cpp \
    KPTListWidget.cpp \
    KPTMenuBar.cpp \
    KPTPushButton.cpp \
    KPTStackedWidget.cpp \
    KPTTextBrowser.cpp


HEADERS += \
    KGuiFw.h \
    KGuiFw_shared.h \
    KPTCheckBox.h \
    KPTPushButton.h \
    KPTFileDialog.h \
    KPTFrame.h \
    KPTLabel.h \
    KPTLineEdit.h \
    KPTListWidget.h \
    KPTMenuBar.h \
    KPTStackedWidget.h \
    KPTTextBrowser.h


FORMS +=

shared: macx: CONFIG += lib_bundle
shared: win32: RC_FILE = app.rc
shared: win32: RESOURCES += res.qrc

OTHER_FILES += app.rc \
    KGuiFw \
    KPTCheckBox \
    KPTFileDialog \
    KPTFrame \
    KPTLabel \
    KPTLineEdit \
    KPTListWidget \
    KPTMenuBar \
    KPTPushButton \
    KPTStackedWidget \
    KPTTextBrowser


unix: !macx: includefiles.path = /opt/keppemessenger/include
unix: !macx: includefiles.files += $$PWD/../include/KCheckBox.h \
    $$PWD/../include/KPTFileDialog.h \
    $$PWD/../include/KPTFrame.h \
    $$PWD/../include/KGuiFw.h \
    $$PWD/../include/KGuiFw_shared.h \
    $$PWD/../include/KPTLabel.h \
    $$PWD/../include/KPTLineEdit.h \
    $$PWD/../include/KPTListWidget.h \
    $$PWD/../include/KPTMenuBar.h \
    $$PWD/../include/KPTPushButton.h \
    $$PWD/../include/KPTStackedWidget.h \
    $$PWD/../include/KPTTextBrowser.h \
    $$PWD/../include/KPTCheckBox \
    $$PWD/../include/KPTFileDialog \
    $$PWD/../include/KPTFrame \
    $$PWD/../include/KGuiFw \
    $$PWD/../include/KPTLabel \
    $$PWD/../include/KPTLineEdit \
    $$PWD/../include/KPTListWidget \
    $$PWD/../include/KPTMenuBar \
    $$PWD/../include/KPTPushButton \
    $$PWD/../include/KPTStackedWidget \
    $$PWD/../include/KPTTextBrowser

unix: !macx: target.path = /opt/keppemessenger/lib
unix: !macx: INSTALLS += target includefiles
